Master in Computational Logics & Mestrado em Engenharia Informática
Universidade Nova de Lisboa

Knowledge Representation and Reasoning

Second Test – Closed book – 2h00m

12th December 2017

Group 1 [7 val.]

Consider the following logic program:

$$\begin{array}{ll} man(X) \leftarrow student(X), not\ woman(X). & student(terry). \\ woman(X) \leftarrow student(X), not\ man(X). & student(john). \quad man(john). \\ human(X) \leftarrow woman(X). & \\ human(X) \leftarrow man(X). & \end{array}$$

1) What is the well-founded model of this program?

Answer: The well founded model can be computed by constructing the sequence

$$T_0 = \{\}$$
$$T_{i+1} = \Gamma_P^2(T_i) = \Gamma_P(\Gamma_P(T_i))$$
$$T_\delta = \cup_{\alpha<\delta} T_\alpha$$

where $\Gamma_P(I) = least(P/I)$, until it reaches a fix-point T, which coincides with the atoms true in the well founded model. Then, the set of atoms false in the well founded model is $F = H_P - \Gamma_P(T)$ (where H_P is the Herbrad Base). We obtain:

$$T = \{man(john), student(john), human(john), student(terry)\}$$
$$F = \{woman(john)\}$$

So, the well founded model is
$M = T \cup not\ F = \{man(john), student(john), human(john), student(terry), not\ woman(john)\}$

2) Using the SLX proof procedure, show whether $human(terry)$ is true (or not).

Answer: Left as exercise.

3) What are the stable models of this program?

Answer: The two stable models are:

$M_1 = \{man(john), student(john), human(john), man(terry), student(terry), human(terry)\}$
$M_2 = \{man(john), student(john), human(john), woman(terry), student(terry), human(terry)\}$

To verify that M_1 is indeed a stable model, we need to determine whether $M_1 = Cn(P^{M_1})$. We have:

$$\begin{aligned} man(terry) &\leftarrow student(terry). \\ man(john) &\leftarrow student(john). \\ human(terry) &\leftarrow woman(terry). \\ human(terry) &\leftarrow man(terry). \\ human(john) &\leftarrow woman(john). \\ human(john) &\leftarrow man(john). \\ student(terry). &\leftarrow \\ student(john). &\leftarrow \\ man(john). &\leftarrow \end{aligned}$$

Since $Cn(P^{M_1}) = \{man(john), student(john), human(john), man(terry), student(terry), human(terry)\} = M_1$ we can conclude that M_1 is a Stable Model. The justification for M_2 is similar.

4) Comment on the differences of the results obtained by the stable model semantics and those by the well-founded semantics in this example. How are these results affected if we add the following rule and fact to the program?

$$student(X) \leftarrow not\ student(X). \qquad woman(mary).$$

Answer: The set of true atoms in the well founded model is a subset of the intersection of all stable models. The set of false atoms in the well founded model are all false in every stable model. There are nevertheless atoms that are true in every stable model that do not belong to the well founded model i.e. the well founded model can be seen as a sound approximation of the intersection of all stable models. However, this is only the case when there is at least one stable model. With the introduction of this rule and fact we obtain the following well founded model:

$$\left\{ \begin{array}{c} man(john), student(john), human(john), student(terry), \\ woman(mary), human(mary), not\ woman(john) \end{array} \right\}$$

However, because of the rule

$$student(mary) \leftarrow not\ student(mary).$$

but we no longer have any stable models. Whereas in some scenarios the absence of stable models can be seen as problematic – in this example there doesn't seem to be any reason not to conclude what belongs to the well founded model – in other scenarios the possibility of the absence of stable models is actually desirable, when such absence indicates that the problem encoded by the logic program doesn't have solutions (as in answer-set programming).

Group 2 [5 val.]

A magic square is a $n \times n$ square grid (where n is the number of cells on each side) filled with distinct positive integers in the range $1, 2, ..., n^2$ such that each cell contains a different integer and the sum of the integers in each row, column and diagonal is equal. The sum is called the magic constant or magic sum of the magic square. A square grid with n cells on each side is said to have order n.



Input Format A particular instance of this problem is described by the number of cells on each side, n, and, to facilitate, also by the definition of the magic constant s:

```
#const n = 3.                % the size
#const s = n*(n*n + 1) / 2. % the magic constant
```

Output Format The output is an assignment of a number `N` to each cell `Row,Col`, encoded as a predicate `x(Row, Col, N)`.

Write an Answer Set Program whose answer sets correspond to the solutions of the problem.

Answer:
```
#const n = 3. % the size

#const s = n*(n*n + 1) / 2. % the magic constant

size(1..n). % facts for the size
val(1..n*n). % facts for the numbers

% exactly one number per cell
1{x(Row,Col,Val):val(Val)}1 :- size(Col), size(Row).

% every number is different
1{x(Row,Col,Val):size(Col), size(Row)}1 :- val(Val).

% sum rows
:- not s #sum{ Val: x(Row, Col, Val), size(Col), val(Val) } s, size(Row).

% sum columns
:- not s #sum{Val: x(Row, Col, Val), size(Row), val(Val) } s, size(Col).

% Sum each diagonal
:- not s #sum { Val:x(I, I, Val), size(I), val(Val) } s.
:- not s #sum { Val:x(I, n-I+1, Val), size(I), val(Val) } s.
```

Group 3 [3 val.]

Contrast, in a clear and concise manner, *First-Order Entailment*, *Entailment with the Closed World Assumption*, and *Minimal Entailment* (aka. *Circumscription*), illustrating with concrete examples.

Answer: See slides...

Group 4 [5 val.]

Someone in Dreadsbury Mansion killed Aunt Agatha. Agatha, the butler, and Charles live in Dreadsbury Mansion, and are the only ones to live there. A killer always hates his victim, and is no richer than his victim. Charles hates no one that Agatha hates. Agatha hates everybody except the butler. The butler hates everyone not richer than Aunt Agatha. The butler hates everyone whom Agatha hates. No one hates everyone. Who killed Agatha?

Write an Answer Set Program to solve this riddle.

Answer:
```
% The people in this drama
person(agatha;butler;charles).

% Exactly one person killed agatha
1{killed(P,agatha):person(P)}1.

% For any two people, only one can be richer that the other one.
{richer(P1,P2);richer(P2,P1)}1 :- person(P1), person(P2), P1!=P2.

% "richer" is a transitive relation.
richer(P1,P3) :- person(P1), person(P2), person(P3), richer(P1,P2), richer(P2,P3).

% everyone can hate everyone.
{hates(P1,P2)} :- person(P1), person(P2).

% A killer always hates her victim
hates(Killer,Victim) :- person(Killer), person(Victim), killed(Killer,Victim).

% and is never richer than her victim
:- richer(Killer,Victim), person(Killer), person(Victim), killed(Killer,Victim).

% Charles hates no one that aunt Agatha hates
0 { hates(charles,X) } 0 :- person(X), hates(agatha,X).

% Agatha hates everyone except the butler, ...
hates(agatha,X) :- person(X), X != butler.

% the butler hates everyone not richer than aunt Agatha
hates(butler, X) :- person(X), not richer(X,agatha).

%The butler hates everyone that agatha hates
hates(butler, X) :- person(X), hates(agatha,X).

% No one hates everyone
:- person(P), {not hates(P,P2) : person(P2)}0.
```

Group 5 [Bonus: 2 val.]

Consider the following program P with aggregates:

$$P = \left\{ \begin{array}{l} p \leftarrow sum\{2:q, 1:s\} \neq 2. \\ q \leftarrow sum\{1:p, 2:s\} \neq 2. \\ \{s\}. \end{array} \right\}$$

What are the stable models of P?

Answer:
$\{p, q, s\}$ and $\{q\}$